

Listing 1

```

10 REM*****
20 REM      Sabiranje realnih brojeva dužine do 255 cifara *
30 REM                                           *
40 REM      Dejan Smiljanić - 1988 *
50 REM*****
60 MODE 2
70 sabirak=0
80 GOSUB 610:REM                               Unos prvog broja
90 zna=zn:aceo=ceo:a$=z$:z$=""
100 GOSUB 610:REM                               Unos drugog broja
110 znb=zn:bceo=ceo:b$=z$:z$=""
120 REM      Izjednačavanje dužine stringova
130 razdec=(LEN(a$)-aceo)-(LEN(b$)-bceo)
140 IF razdec>0 THEN b$=b$+STRING$(razdec,"0")
150 IF razdec<0 THEN a$=a$+STRING$(-razdec,"0")
160 razceo=aceo-bceo
170 IF razceo>=0 THEN b$=STRING$(razceo,"0")+b$:
    bceo=bceo+razceo
180 IF razceo<0 THEN a$=STRING$(-razceo,"0")+a$:
    aceo=aceo-razceo
190 REM      Utvrđivanje apsolutno većeg broja
200 IF zna*znb=1 THEN 260
210 FOR i=1 TO LEN(a$)
220 IF MID$(a$,i,1)=". " OR MID$(b$,i,1)=". " THEN 250
230 ab=VAL(MID$(a$,i,1))-VAL(MID$(b$,i,1))
240 IF ab<>0 THEN 260
250 NEXT i
260 DIM z(LEN(a$)):REM                               Sabiranje,cifru po cifru
270 FOR i=LEN(a$) TO 1 STEP -1
280 IF MID$(a$,i,1)=". " THEN z(i-1)=z(i):GOTO 340
290 IF zna*znb=1 THEN z(i)=z(i)+VAL(MID$(a$,i,1))+
    VAL(MID$(b$,i,1))
300 IF zna*znb=-1 AND ab>0 THEN z(i)=z(i)+VAL(MID$(a$,i,1))
    -VAL(MID$(b$,i,1))
310 IF zna*znb=-1 AND ab<0 THEN z(i)=z(i)+VAL(MID$(b$,i,1))
    -VAL(MID$(a$,i,1))
320 IF z(i)>=10 THEN z(i)=z(i)-10:z(i-1)=z(i-1)+1
330 IF z(i)<0 THEN z(i)=z(i)+10:z(i-1)=z(i-1)-1
340 NEXT i
350 REM      Formiranje zbira i preslikavanje u prvi broj
360 nul=0:duza=LEN(a$):a$="":b$="":cceo=0
370 FOR i=0 TO aceo-1
380 IF RIGHT$(STR$(z(i)),1)<>"0" THEN nul=1
390 IF nul=1 THEN c$=c$+RIGHT$(STR$(z(i)),1):
    a$=a$+RIGHT$(STR$(z(i)),1):cceo=cceo+1
400 NEXT i
410 c$=c$+ ". ":a$=a$+ ". ":cceo=cceo+1
420 FOR i=aceo+1 TO duza
430 c$=c$+RIGHT$(STR$(z(i)),1)
440 a$=a$+RIGHT$(STR$(z(i)),1)
450 NEXT i
460 IF LEFT$(c$,1)=". " THEN c$="0"+c$
470 IF RIGHT$(c$,1)=". " THEN c$=LEFT$(c$,LEN(c$)-1)
480 znc=1
490 IF zna=-1 AND znb=-1 THEN c$="-"+c$:znc=-1
500 IF zna=-1 AND znb=1 AND ab>0 THEN c$="-"+c$:znc=-1
510 IF zna=1 AND znb=-1 AND ab<0 THEN c$="-"+c$:znc=-1
520 REM      Anuliranje pomoćnih promenljivih i ispis zbira
530 ERASE z:zna=znb:aceo=cceo:razdec=0:razceo=0:ab=0:nul=0
540 PRINT".....Tekući zbir: ";c$:c$="":GOTO 100
600 REM      Podprogram za unos brojeva
610 tacka=0:zn=1:ceo=0:sabirak=sabirak+1:
    PRINT sabirak;"sabirak:";
620 INPUT z$
630 IF z$="" THEN 620
640 IF LEFT$(z$,1)="+ " THEN z$=RIGHT$(z$,LEN(z$)-1)
650 IF LEFT$(z$,1)="- " THEN zn=-1:z$=RIGHT$(z$,LEN(z$)-1)
660 FOR i=1 TO LEN(z$)
670 IF ASC(MID$(z$,i,1))=46 AND tacka=1 THEN 730
680 IF ASC(MID$(z$,i,1))=46 THEN tacka=1:ceo=i
690 IF ASC(MID$(z$,i,1))<46 OR ASC(MID$(z$,i,1))>57 OR
    ASC(MID$(z$,i,1))=47 THEN 730
700 NEXT i
710 IF ceo=0 THEN z$=z$+ ". ":tacka=1:ceo=LEN(z$)
720 RETURN
730 PRINT"Pogrešan unos - ponovi!";CHR$(7):
    sabirak=sabirak-1:GOTO 610

```

Listing 2

```
600 REM                                     Podprogram za unos brojeva
610 tacka=0:zn=0:ceo=0:sabirak=sabirak+1:
    PRINT sabirak;"sabirak:";
620 IF zn<>0 THEN 660
630 IF INKEY(28)=32 THEN zn=1:PRINT"+";
640 IF INKEY(25)=0 THEN zn=-1:PRINT"-";
650 GOTO 620
660 i%=1
670 IF tacka=1 THEN 690
680 IF INKEY(31)=0 OR INKEY(7)=0 THEN tacka=1:PRINT".":
    z$=z$+"." :ceo=i%:i%=i%+1
690 RESTORE 800
700 FOR n%=1 TO 10
710 READ taster%,tipka%
720 IF INKEY(taster%)=0 OR INKEY(tipka%)=0 THEN PRINT
    CHR$(47+n%);:z$=z$+CHR$(47+n%):i%=i%+1
730 IF INKEY(taster%)=0 OR INKEY(tipka%)=0 THEN CALL &BB03:
    GOTO 730
740 NEXT n%
750 IF INKEY(66)=0 THEN END
760 IF (INKEY(18)=0 OR INKEY(6)=0) AND tacka=1 THEN PRINT"":
    GOTO 790
770 IF (INKEY(18)=0 OR INKEY(6)=0) AND tacka=0 THEN PRINT"":
    tacka=1:z$=z$+"." :ceo=LEN(z$):GOTO 790
780 GOTO 670
790 RETURN
800 DATA 32,15,64,13,65,14,57,5,56,20
810 DATA 49,12,48,4,41,10,40,11,33,3
```

Listing 3

```
600 REM                                     Podprogram za unos brojeva
610 tacka=0:zn=0:ceo=0:sabirak=sabirak+1:
    PRINT sabirak;"sabirak:";
620 IF zn<>0 THEN 670
630 tast$=INKEY$
640 IF tast$="+" THEN zn=1:PRINT"+";
650 IF tast$="-" THEN zn=-1:PRINT"-";
660 GOTO 620
670 i%=1
680 tast$=INKEY$
690 IF tacka=1 THEN 710
700 IF tast$="." THEN tacka=1:PRINT".":z$=z$+"." :ceo=i%:
    i%=i%+1
710 FOR n%=0 TO 9
720 IF tast$=RIGHT$(STR$(n%),1) THEN PRINT tast$::
    z$=z$+tast$:i%=i%+1
730 NEXT n%
740 IF tast$=CHR$(13) AND tacka=1 THEN PRINT"":GOTO 770
750 IF tast$=CHR$(13) AND tacka=0 THEN PRINT"":tacka=1:
    z$=z$+"." :ceo=LEN(z$):GOTO 770
760 GOTO 680
770 RETURN
```